



# Getting Started with the MapControl

**GIS by ESRI™**

Copyright © 2000–2002 ESRI

All Rights Reserved.

Printed in the United States of America.

The information contained in this document is the exclusive property of ESRI. This work is protected under United States copyright law and the copyright laws of the given countries of origin and applicable international laws, treaties, and/or conventions. No part of this work may be reproduced or transmitted in any form or by any means, electronic or mechanical, including photocopying or recording, or by any information storage or retrieval system, except as expressly permitted in writing by ESRI. All requests should be sent to Attention: Contracts Manager, ESRI, 380 New York Street, Redlands, CA 92373-8100 USA.

The information contained in this document is subject to change without notice.

### **SAMPLE DATA LICENSE AND DISCLAIMER**

The sample data included in this package is the exclusive property and copyright of ESRI and the respective data publishers. This sample data is protected under United States copyright law and other international copyright treaties and conventions. The sample data is provided under license from each of the respective data publishers for the Licensee's own internal use. Licensee shall not sell, rent, lease, sublicense, lend, assign, time-share, or transfer, in whole or in part, or provide unlicensed third parties access to the sample data or portions of the data, documentation or meta data, any updates, or Licensee's rights under this Agreement. The sample data herein have been obtained from sources believed to be reliable, but its accuracy and completeness, and the opinions based thereon, are not guaranteed. Every effort has been made to provide accurate sample data in this package. ESRI is not inviting reliance on this sample data, and you should always verify actual map data and information. The sample data contained in this package is subject to change without notice. THE SAMPLE DATA IS PROVIDED "AS-IS," WITHOUT WARRANTY OF ANY KIND, EITHER EXPRESS OR IMPLIED, INCLUDING BUT NOT LIMITED TO, THE IMPLIED WARRANTIES OF MERCHANTABILITY AND FITNESS FOR A PARTICULAR PURPOSE. ESRI shall not be liable for indirect, special, incidental, or consequential damages related to Licensee's use of the sample data, even if ESRI is advised of the possibility of such damage.

### **RESTRICTED RIGHTS LEGEND**

Use, duplication, and disclosure by the U.S. Government are subject to restrictions as set forth in FAR §164§52.227-14 Alternate III (g)(3) (JUN 1987), FAR §164§52.227-19 (JUN 1987), DFARS §164§252.227-7015 (JUN 1995) [Technical Data], and/or DFARS §164§252.227-7202 (JUN 1995) [Computer Software], as applicable. Contractor/Manufacturer is ESRI, 380 New York Street, Redlands, CA 92373-8100 USA.

ESRI, MapControl, ArcMap, Spatial Database Engine, ArcSDE, the ESRI corporate logo, the ArcObjects logo, and ArcGIS by ESRI are trademarks, registered trademarks, or service marks of ESRI in the United States, the European Community, or certain other jurisdictions.

The names of other companies and products herein are trademarks or registered trademarks of their respective trademark owners.

# Getting Started with the MapControl

## IN THIS BOOK

- Loading the MapControl
- Creating a Visual Basic template
- Adding a MapControl
- Displaying property pages and obtaining help
- Using the property pages
- Updating properties and editing code
- Using the ArcObjects Developer Help
- Adding map navigation functionality
- Handling form resize
- Displaying map layers based on a scale
- Adding layers through code
- ArcObjects Controls Resources

The MapControl is one of the ArcObjects ActiveX Controls. When embedded within a development environment, it provides a window similar to data view in ArcMap. In this book, you will use the MapControl in conjunction with Microsoft Visual Basic version 6. You will learn how to:

- Display a map with multiple layers
- Add and manage data
- Display features with thematic renderers
- Control the general appearance of the control
- Set map properties
- Control panning and zooming
- Display map layers based on scale
- Perform logical queries to find features

# Loading the MapControl

MapControl 8.3 is contained within the ESRI MapControl 8.3 library. The filename for this control is MapControl.ocx.


ArcObjects and functions are contained within the ESRI Object Library (esriCore.olb). Supplementary objects and functions are contained within the ESRI Controls Support Library 8.3 (controlsSupport.dll).

1. Start Visual Basic and create a new Standard EXE project from the New Project dialog box.
2. Click the Project menu and click Components.
3. In the Components dialog box, click ESRI MapControl 8.3 and click OK.

The MapControl appears in the Visual Basic toolbox.

4. Click the Project menu again and click References.
5. In the References dialog box, click ESRI Object Library and ESRI Controls Support Library 8.3, then click OK.

If ESRI Controls Support Library is not available in the list, click the Browse button, navigate to the Bin folder of your ArcGIS installation, and select the ControlsSupport.dll.



MapControl icon

## Creating a Visual Basic template

When you open a new project in Visual Basic (VB), your toolbox contains some, but not all, of the custom controls available with your version of VB. You might find it convenient to tailor the default controls that are loaded. You can do this by creating a project template of the controls you wish to have loaded.

If you accepted the default when installing VB, the template folder to use is C:\Program Files\Microsoft Visual Studio\VB98\Template\Projects.

You can specify a template project in VB which contains references to all the ArcObjects controls. You can then use this project as a template whenever you create a new application using the ArcObjects controls. For more details, see the following sample in the ArcObjects Developer Help: \Samples\Controls\AllControls\VBTemplate.


## Creating a template for the MapControl

1. Follow steps 1–5 in the 'Loading the MapControl' section to load the MapControl into a new Visual Basic project.
2. Click the File menu and click Save Project As.
3. Navigate to the Visual Basic template folder C:\Program Files\Microsoft Visual Studio\VB98\Template\Projects and type the filename ArcGIS Controls.vbp.
4. Click Save.

## Using the MapControl template

1. Open Visual Basic and open the template you created from the New Project dialog box.

Visual Basic opens with the MapControl already loaded into the toolbox.




## Adding a MapControl

You can add one or more MapControls to any Visual Basic form.

When you add a MapControl, it is initially white with its name in the center. You have to set its properties.

1. Click the MapControl icon in the Visual Basic toolbox.
2. Click and drag from one corner to the opposite corner on a Visual Basic form to place the MapControl.

Alternatively, double-click the MapControl icon in the Visual Basic toolbox and the MapControl will automatically be placed on the form.





# Displaying property pages and obtaining help

The MapControl has five property pages: General, Layers, Map, Color, and Picture.

Use the General property page to set the properties controlling the general appearance of the MapControl.

Use the Layers property page to add and manage data layers to be loaded into the MapControl.

Use the Map property page to set the properties that refer to the map within the MapControl.

You can set the background color of the MapControl using the Color property page.

The Picture property page lets you specify a custom mouse pointer icon. This icon is used when the mouse pointer is over the map control. If you use a custom icon, you must also set the MousePointer property to esriPointerCustom—you can do this either programmatically or on the General property page.

## Displaying the property pages

1. Right-click the MapControl and click Properties to open the MapControl property pages.




## Obtaining property page help using the What's This button

1. Right-click over a control or label on the Property Pages dialog box.
2. Click the What's This button. Help for the control displays.
3. Click anywhere on the Property Pages dialog box to close the help.

## Obtaining property page help using the Help button

4. Click the Help button on the Property Pages dialog box.

The ArcObjects Developer Help will open and display help about the selected property page.



## Using the property pages to add data

A map document can be an ArcMap document or template (.mxd, .mxt), or a published map file (.pmf). If you choose to load data from a .pmf, you may be asked to enter a password.

The sample documents can be found on the ArcGIS Digital Books and Sample Maps CD.




Using the Load Selected Map button will reset many existing properties of the Map Control, such as Spatial Reference, MapScale, and Rotation. Any graphics or text elements within the map will be included.

## Adding data layers

1. Right-click the MapControl and click Properties to open the MapControl property pages.
2. Click the Layers tab.
3. Click the Add Layers button.
4. In the Add Layers dialog box, navigate to a folder containing sample data.
5. Click on and highlight the data you want to add. To highlight multiple data layers, hold down the Shift key. Then, click the Add Layers button.
6. Click OK to apply the changes you have made and close the property pages.
7. Click the Run menu and click Start.



## Adding data from map documents

1. Right-click the MapControl and click Properties to open the MapControl property pages.
2. Click the Layers tab. ▶







3. Click the Browse Map Document button.
4. In the Map Documents dialog box, navigate to a folder containing a map document of your choice.
5. Click on and highlight a map document, then click the Open button.
6. Select a map in the Maps in the Map Document dropdown list from which to add data.
7. Click the Load Selected Map button to add all data layers within the selected map.
8. Select a specific layer within the Layers in Selected Map box. Click the Load Selected Layer button to add the selected layer. To select multiple data layers, hold down the Shift key.
9. Optionally, repeat step 6 to select another map from which to add data.
10. Click OK to apply the changes you have made and close the property pages.
11. Click the Run menu and click Start.



## Using the property pages to manage data

Typically, raster data layers should be moved down the layer list and the drawing order of the MapControl. Other vector data layers will then be visible on top of the raster data.

1. Right-click the MapControl and click Properties to open the MapControl property pages.
2. Click the Layers tab.
3. Assuming that data has already been added, select and highlight a layer in the MapControl Layers list box. If data has not been added, see the task 'Adding data from map documents'.
4. Click the Move Layer Up and Move Layer Down buttons to move the selected layer up or down the layer list and the drawing order for the MapControl.
5. Click the Properties button to display the Layer Properties dialog box for the selected layer. Use the dialog box to change any layer properties, such as symbology.
6. Click the Delete Layer button to remove the selected layer from the MapControl Layers list.



## Setting control appearance with the property pages




1. Right-click the MapControl and click Properties to open the MapControl property pages.
2. Click the General tab.
3. Check the Preview in Design Mode check box.
4. Click Apply.

If necessary, move the property pages so that the MapControl can be seen on the form. The MapControl will appear on the form as it would at run time.

5. Optionally, select a different Border Style or Appearance from the dropdown lists.
6. Optionally, select a different Mouse Pointer from the Mouse Pointer dropdown list.

The selected Mouse Pointer will display as the mouse passes over the control.




7. Click Apply to view the changes while in design mode.
8. Click OK to apply the changes you have made and close the property pages.
9. Click the Run menu and click Start.



## Using the property pages to set map properties

The Map property page allows the setting of properties related to the coordinate system and scale of the MapControl's map.

1. Add data to the control (see the section 'Using the MapControl property pages to add data').
2. Click the Run menu and click Start.  
View the data in the MapControl.
3. Click the Run menu and click End.
4. Right-click the MapControl and click Properties to open the MapControl property pages.
5. Click the Map tab.
6. In the Rotation (degrees) text box, type in a rotation between 0 and 360.
7. Click OK to apply the changes you have made and close the property pages.
8. Click the Run menu and click Start.



# Updating properties and editing code

The Visual Basic property window and the MapControl property pages allow easy access to some MapControl properties without the need to write code.

The subset of properties available via the Visual Basic property window are limited to those that have simple set or return types, such as integers and strings. These properties are from both the IMapControl2 interface (for example, MapUnits or ReferenceScale) and the set of standard Visual Basic control properties (for example, Name, Height, and Width).

The MapControl property pages allow access to a subset of properties taken only from the IMapControl2 interface, but include some properties with more complex set or return types, such as Layers and Spatial Reference.

## Updating properties

1. Click the MapControl on the form to make it active.
2. Press F4 or click the View menu and click Properties Window.
3. Click the (...) button by the Custom property to open the MapControl's property pages.



## Editing Code

1. Click the MapControl on the form to make it active.
2. Click F7, double-click the MapControl, or click the View menu and click Code.

The OnMouseDown event is the MapControl's default event.

3. Click the dropdown arrow on the right-hand side and click an event to edit.

Visual Basic creates code stubs for each event so that you can immediately begin adding code.



# Using the ArcObjects Developer Help

The ArcObjects Developer Help contains documentation on every object, property, method, and event within ArcObjects, including the ArcGIS Controls.

The ArcObjects Developer Help has a section devoted to the ArcObjects Controls. To access it, click the Contents tab and open the ArcObjects Controls book.

## Opening ArcObjects Developer Help




1. Make the MapControl active by clicking it.
2. Press F1.

## Getting help through the Object Browser

1. Press F2 or click the View menu and choose Object Browser.
2. Select esriMapControl from the Project/Library dropdown list.
3. Click MapControl in the Classes list box and click an event, method, or property from the Members list.
4. Click the Help button to open help about the event, method, or property you are interested in.

## Getting help from the edit window

1. Make the MapControl active by clicking it.
2. Press F7, double-click the MapControl, or click the View menu and click Code.
3. Hover the pointer over a word that is an ArcObjects object, property, method, or event.
4. Press F1.



## Adding map navigation functionality

At this point, the MapControl can display data at its full extent. By writing code in the OnMouseDown event, simple pan and zoom functionality can be implemented.

The TrackRectangle method applies to the MapControl. It tracks the movement of the mouse while the user keeps a mouse button pressed and at the same time rubberbands a rectangle. When the user releases the mouse button, the TrackRectangle method returns an IEnvelope object that the application assigns to the Extent property of the map, causing the map to be redrawn at the new extent. ►

## Adding ZoomIn functionality

1. Double-click the MapControl to display the Visual Basic code window.

2. Add the following code to the OnMouseDown procedure:

```
Private Sub MapControl1_OnMouseDown(ByVal button As Long, ByVal shift _  
As Long, ByVal x As Long, ByVal y As Long, ByVal mapX As Double, ByVal _  
mapY As Double)
```

```
    MapControl1.Extent = MapControl1.TrackRectangle
```

```
End Sub
```



3. Click the Run menu and click Start.

4. Click the map with the left mouse button and drag out a rectangle.

5. Release the mouse button.

The map redraws.

6. Click the Run menu and click Stop to return to design mode.





If the left mouse button is pressed, the `OnMouseDown` event is passed a `button` parameter equal to `vbLeftButton` and the code to zoom in will execute. If the right mouse button is pressed, a `button` parameter equal to `vbRightButton` is passed and the code will execute the `Pan` method. ▶

## Adding Pan functionality

1. Double-click the `MapControl` to display the Visual Basic code window.

2. Amend the code in the `OnMouseDown` procedure as follows:

```
Private Sub MapControl1_OnMouseDown(ByVal button As Long, ByVal shift _  
As Long, ByVal x As Long, ByVal y As Long, ByVal mapX As Double, ByVal _  
mapY As Double)
```

```
    If button = vbLeftButton Then
```

```
        MapControl1.Extent = MapControl1.TrackRectangle
```

```
    ElseIf button = vbRightButton Then
```

```
        MapControl1.Pan
```

```
    End If
```

```
End Sub
```



3. Click the `Run` menu and click `Start`.

4. Click and drag the map to a new location.

5. Release the mouse button.

The map redraws.

6. Click the `Run` menu and click `Stop` to return to design mode.




The MapControl FullExtent property returns an IEnvelope object that defines a bounding box around all the layers in the map.

## Adding Zoom to Full Extent functionality

1. Double-click the CommandButton button in the toolbox to add a command button to the form.
2. Click the command button and drag it to the lower-right corner of the form.
3. Press F4 to display the properties window.
4. Click in the caption box and type Full Extent to change the button's caption.
5. Double-click the Full Extent button to display the code window.
6. Add the following code to the Click procedure:

```
Private Sub Command1_Click()  
    MapControl1.Extent = MapControl1.FullExtent  
End Sub
```

7. Click the MapControl and press F4 to display the properties window. Set the ShowScrollbars property to False.
8. Click the Run menu and click Start.
9. Use the mouse buttons to zoom in on and pan around the data.
10. Click the Full Extent button to redraw the map at the full extent.
11. Click the Run menu and click Stop to return to design mode.



## Creating a Find tool

In this section, you will add additional controls to your application to implement a simple function for locating a state by name and zooming to it.

The sample data used in this example can be found in the following location of your ArcGIS installation:

```
\\arcgis\arcexe82\ArcObjects  
Developer Kit\  
Samples\Data\Usa
```

The code locates the states layer and defines a simple SQL query expression using the text in the text box. The states layer's IFeatureSelection uses this query to search for the specified value and return the first matching record. The extent of the MapControl is set to the extent of the selected feature's geometry.

1. Add the states.shp and ushigh.shp sample datasets to the MapControl using the property pages. For more information about how to do this, see the 'Using the property pages to add data' section.
2. Double-click the Label button in the toolbox to add a label to the form.
3. Click the label and drag it to the lower-left corner of the form.
4. Press F4 to display the properties window, then set the caption of the label to State.
5. Double-click the TextBox button in the toolbox to add a text box to the form.
6. Click and drag the text box to position it next to the label.
7. Press F4 to display the properties window, then clear the text property of the text box.
8. Double-click the TextBox to display the code window.
9. Select the Text1\_KeyPress event from the right-hand dropdown list.
10. Add the following code to the Text1\_KeyPress procedure:


```
Private Sub Text1_KeyPress(KeyAscii As Integer)  
    'If the return key has been pressed  
    If KeyAscii = vbKeyReturn Then  
        Dim i As Integer  
        Dim pFeatureLayer As IFeatureLayer  
        'Loop through the MapControl layers to find the states layer  
        For i = 0 To MapControl1.LayerCount - 1  
            Set pFeatureLayer = MapControl1.Layer(i)  
            If pFeatureLayer.Name = "states" Then  
                Exit For  
            End If  
        Next i  
        'Create a query string  
        Dim sQuery As String  
        sQuery = "STATE_NAME = "" & Text1.Text & ""  
        'Create the query filter  
        Dim pQueryFilter As IQueryFilter  
        Set pQueryFilter = New QueryFilter  
        pQueryFilter.WhereClause = sQuery ▶  
        'Query interface for IFeatureSelection and perform selection
```

```

Dim pFeatureSelection As IFeatureSelection
Set pFeatureSelection = pFeatureLayer
pFeatureSelection.SelectFeatures pQueryFilter, _
esriSelectionResultNew, True
'Get the selection set
Dim pSelectionSet As ISelectionSet
Set pSelectionSet = pFeatureSelection.SelectionSet
'Get the feature cursor
Dim pFeatureCursor As IFeatureCursor
pSelectionSet.Search Nothing, True, pFeatureCursor
'Get the selected feature
Dim pFeature As IFeature
Set pFeature = pFeatureCursor.NextFeature
If pFeature Is Nothing Then Exit Sub
'Set the MapControl extent to be the extent of the selected
feature
MapControl1.Extent = pFeature.Shape.Envelope
End If
End Sub

```

11. Click the Run menu and click Start.
12. Type the name of a state into the TextBox, for example, New Jersey.
13. Press Enter.
14. Click the Run menu and click Stop to return to design mode.



## Handling form resize

When a form is resized, the MapControl does not automatically resize itself. This example resizes the controls using the Move method.

Notice that when you run your application, it initially redraws the map twice. This is due to the fact that controls on the form are initially displayed using the size and position specified during design time. To fix this problem, you will resize the controls when the form is loaded. You have already written the code to resize the controls, you just need to call the procedure.


## Responding to the Form Resize event

1. Double-click the form to display the code window.
2. Select the Resize event from the right-hand dropdown list.
3. Add the following code to the Form Resize procedure:

```
Private Sub Form_Resize()  
    If (ScaleWidth <> 0) Then  
        'Constant spacing  
        Dim iSpace As Integer  
        iSpace = Text1.Top - (MapControl1.Top + MapControl1.Height)  
        'y coordinate for text box, label and command  
        Dim iYCoord As Integer  
        iYCoord = ScaleHeight - Text1.Height - iSpace  
        'MapControl height  
        Dim iMapHeight As Integer  
        iMapHeight = iYCoord - iSpace  
        'Move the controls  
        If (iMapHeight > 0) Then  
            Command1.Move ScaleWidth - Command1.Width, iYCoord  
            Text1.Move Text1.Left, iYCoord  
            Label1.Move Label1.Left, iYCoord  
            MapControl1.Move 0, 0, ScaleWidth, iMapHeight  
        End If  
    End If  
End Sub
```

4. Double-click the form to show the code window.
5. Add the following code to the Form Load procedure:

```
Private Sub Form_Load()  
    Form_Resize  
End Sub
```

- 
1. Click the MapControl and press F4 to display the properties window. Set the ShowScrollbars property to False.
  2. Click the Run menu and click Start.
  3. Use the mouse buttons to zoom in on and pan around the data.
  4. Click the Full Extent button to redraw the map at the full extent.
  5. Click the Run menu and click Stop to return to design mode.

## Displaying map layers based on scale

In this section, a new layer is added to the MapControl and code is added to determine whether or not the layer is visible. At certain scales there is no need to show too much detail.

The sample data used in this example can be found in the following location of your ArcGIS installation:

```
\\arcgis\arcexe83\ArcObjects  
Developer  
Kit\Samples\Data\Usa
```



The ILayer MinimumScale and MaximumScale properties are used to selectively make the counties and states layers visible or invisible depending on the map scale.

## Setting scale breaks

1. Add the counties.shp to the MapControl using the property pages. For more information about how to do this, see the 'Using the property pages to add data' section.
2. Click the counties layer in the property page to select it, then use the arrow buttons to move it to the middle of the layer list (beneath the ushigh layer).
3. Click the Properties button to display the counties layer's properties. Click the Symbology tab, click Categories, then click Unique Values.
4. Click STATE\_NAME in the Value Field and click the Add All Values button, then click OK in the Layer Properties dialog box and click OK in the property pages.
5. Add the following code to the Form Load procedure (after Form Resize):

```
Dim pLayer As ILayer  
Dim i As Integer  
'Loop through the MapControl layers  
For i = 0 To MapControl1.LayerCount - 1  
    Set pLayer = MapControl1.Layer(i)  
    'Set scale thresholds  
    If UCase(pLayer.Name) = "COUNTIES" Then '(note case sensitivity)  
        pLayer.MaximumScale = 0#  
        pLayer.MinimumScale = 15000000#  
    ElseIf UCase(pLayer.Name) = "STATES" Then '(note case sensitivity)  
        pLayer.MaximumScale = 14999999#  
        pLayer.MinimumScale = 0#  
    End If  
Next i
```

6. Run the application. Zoom in until the counties layer becomes visible. Notice that when you click the Full Extent button, the counties layer is no longer visible.





## Adding layers through code

In the previous sections, the layers have been specified interactively using the MapControl's property page. In this section, code is added to create Layer objects programmatically.

The sample data used in this example can be found in the following location of your ArcGIS installation:

```
\\arcgis\arcexe83\ArcObjects  
Developer  
Kit\Samples\Data\Usa
```

The AddShapefile method adds a new layer with default symbology, specified by its path and filename, to the top of any layers in the MapControl.


The AddLayer method takes an ILayer object and adds it to the MapControl at the specified index. Layers can be created and loaded from different data formats: TIN, Raster, coverage, or ArcSDE.

The LoadMxFile and AddLayerFromFile methods can also be used to add data to the MapControl.

## Loading data programmatically

1. Display the MapControl property pages and click the Layers tab.
2. Click the ushigh layer, then click the Delete button to remove it.
3. Repeat step 2 for the counties and states layers, then click OK to dismiss the property pages.
4. Double-click the form to show the code window.
5. Select the General section from the left-hand dropdown list and add the following procedure to the General section:

```
Private Sub LoadMapData()  
    'Change this path to reflect your ArcGIS installation  
    Dim sPath As String  
    sPath = "D:\arcgis\arcexe83\ArcObjects Developer Kit\Samples\Data\Usa"  
    'Create a new ShapefileworkspaceFactory object  
    Dim pWorkspaceFactory As IWorkspaceFactory  
    Set pWorkspaceFactory = New ShapefileworkspaceFactory  
    'Create a feature workspace  
    Dim pFeatureworkspace As IFeatureworkspace  
    Set pFeatureworkspace = pWorkspaceFactory.OpenFromFile(sPath, 0)  
    'Create a new feature layer and assign it a shapefile  
    Dim pFeatureLayer As IFeatureLayer  
    Set pFeatureLayer = New FeatureLayer  
    Set pFeatureLayer.FeatureClass = _  
    pFeatureworkspace.OpenFeatureClass("states")  
    pFeatureLayer.Name = pFeatureLayer.FeatureClass.AliasName  
    'Add the feature layer to the map  
    MapControl1.AddLayer pFeatureLayer  
    'Add the shapefiles to the map  
    MapControl1.AddShapeFile sPath, "counties"  
    MapControl1.AddShapeFile sPath, "ushigh"  
End Sub ▶
```

- 
6. Insert the following line of code into the Form Load event immediately after the call to `Form_Resize` and before the section of code that sets the minimum and maximum display scales for the layers:

`LoadMapData`

7. Run the application.

The map should appear as before, but default colors and renderers are used that may be different from the colors previously selected in the Layer Properties dialog box.

# ArcObjects Controls Resources

The ArcObjects ActiveX Controls, including the MapControl and PageLayoutControl, come with additional resources to help you get started:

- ArcObjects Developer Help
- Component Help
- Samples
- Object Model Diagrams
- User Forums
- ArcObjects Online

1. Open the ArcObjects Developer Help to obtain more information about the controls and access samples, component help, and object model diagrams.
2. Go to ArcObjects Online (can access it via <http://support.esri.com>) to view an online version of the ArcObjects Developer Help. The online version is frequently updated with new samples and component help.
3. Go to the ArcGIS Discussion Forum: <http://support.esri.com/forums/>

